****************************************************************************
* Nome do Arquivo: 		codigo90.do
* Data:			 		23/03/2020
* Autor: 				Fred Batista
* Prop�sito: 			An�lise da pesquisa DAT/BR90.MAR-00219
* Dados: 				00219.dta
* Output:			-	*/
****************************************************************************

* VARI�VEIS DO QUESTION�RIO NO BANCO

* escala esquerda-direita: q32
* regiao: regiao
* estado: estado
* sexo: sexo
* idade: idade
* avalia��o Sarney = q1 
* voto 1o turno: q6
* voto 2o turno: q7
* partido: partido
* exposi��o � informa��o: q30a-q30d
* le noticiario: q31a
* conversa sore politica: q31b
* trabalha: q40
* catolico: q44a
* escolaridade: escola
* renda: renda
* opiniao sobre o Ҏ melhor manter a familia do que acabar com um relacionamento infeliz�: q35c


* RECODIFICA��O

* opinina��o na escala direita-esquerda

gen loc_resp = q32
recode loc_resp (0 8 9 10=0) (else=1)

* localiza��o na escala direita-esquerda

gen diresq_resp = q32
recode diresq_resp (0 8 9 10=.)
replace diresq_resp = (diresq_resp - 1)/6

* sexo

gen homem = sexo
recode homem (1=1) (2=0)

* avalia��o Sarney

gen avgov = q1
recode avgov (6=.)
replace avgov  = (5 - avgov)/4

* voto (Lula Brizola Freire a esquerda; Covas Guimaraes Afif Camargo ao centro; Collor Maluf Chaves a direita), base em Carreirao 2002

gen voto1 = q6
recode voto1 (4 5 8= 0) (3 6 10 11=1) (1 2 7=2) (else=.)

* voto segundo turno

gen voto2 = q7
recode voto2 (2=1) (1=0) (else=.)

* partidarismo

gen idpmdb = partido
recode idpmdb (1=1) (else=0)

gen idpt = partido
recode idpt (2=1) (else=0)

gen idprn = partido
recode idprn (8=1) (else=0)

* le noticias de politica

gen leitura = q31a
recode leitura (1=1) (0 2 3=0)

* conversa de politica

gen conversa = q31b
recode conversa (1=1) (0 2 3=0)

* trabalha

gen trabalha = q40
recode trabalha (1=1) (0 2 3 4 5 6=0)

* catolico

gen catolico = q44a
recode catolico (1=1) (2 3 4=0)

* escolaridade 

gen escolaridade = (escola - 1)/7

* renda

recode renda (6=.)

*** AN�LISES

* opina��o

probit loc_resp idpmdb idpt idprn escolaridade conversa leitura homem i.renda i.idade i.regiao trabalha catolico

margins, dydx(escolaridade)

* o coeficiente da escolaridade � o mais alto do modelo


* voto

* 1 turno

mprobit voto1 diresq_resp, base(0)

margins, dydx(diresq_resp) predict(outcome(1))

margins, dydx(diresq_resp) predict(outcome(2))

mprobit voto1 diresq_resp avgov idpmdb idpt idprn escolaridade conversa leitura homem i.renda i.idade i.regiao trabalha catolico, base(0)

margins, dydx(diresq_resp) predict(outcome(1))

margins, dydx(diresq_resp) predict(outcome(2))

mprobit voto1 c.diresq_resp##c.escolaridade avgov idpmdb idpt idprn conversa leitura homem i.renda i.idade i.regiao trabalha catolico, base(0)

margins, dydx(diresq_resp) at(escolaridade=(0(1)1)) predict(outcome(1))

margins, dydx(diresq_resp) at(escolaridade=(0(1)1)) predict(outcome(2))


* 2 turno

probit voto2 diresq_resp

margins, dydx(diresq_resp) 

probit voto2 diresq_resp avgov idpmdb idpt idprn escolaridade conversa leitura homem i.renda i.idade i.regiao trabalha catolico

margins, dydx(diresq_resp) 

probit voto2 c.diresq_resp##c.escolaridade avgov idpmdb idpt idprn conversa leitura homem i.renda i.idade i.regiao trabalha catolico

margins, dydx(diresq_resp) at(escolaridade=(0(1)1))
